
Figure 1.  Columns in the SAS adverse event table. 

ABSTRACT 
 
Transferring data between SAS and Microsoft Office can be difficult, especially when SAS is not installed on a 
Windows® platform.  This paper discusses using the HTML and XML support in BASE SAS software to move data 
between SAS and Microsoft Office (versions 2002 and later).  You can use the techniques described here regardless 
of the platform on which SAS software is installed.  

INTRODUCTION 
 
After reading this paper you should be able to import into a Microsoft Excel or Word document most SAS output, as 
well as import Excel workbooks into SAS.  The data and sample SAS code used in this paper reveal formatting 
issues that you may encounter when attempting to import HTML and XML output generated by the Output Delivery 
System (ODS) into Office documents.  The TEMPLATE procedure is used to correct these problems by modifying an 
ODS style.  The paper also explains and provides code for importing Excel workbooks into SAS tables.  All code was 
tested with SAS 9.1 and Office XP Service Pack 3. 
 
For information about generating other types of output and techniques for using those types of output in Excel and 
Word, refer to this author's SUGI 27 paper (DelGobbo, 2002). 
 

SAMPLE SAS DATA 
 
Figure 1 shows the column 
properties of the data that we 
are going to move to Excel and 
Word.  The data is adverse 
event information for a fictitious 
drug. 
 
Note that labels have been 
applied to each column in the 
table.  This allows for a more 
attractive table when imported 
into the Office applications. 
 
The two columns AESEV and 
AESEVC represent the severity 
of the adverse event.  As 
illustrated in Figure 2 (below), 
the same data is represented 
two ways: AESEV contains a 
numeric value for the severity 
while AESEVC contains a 
character value.  The SAS code 
later in this paper makes use of 
both columns. 

Figure 2.  Partial view of the SAS adverse event table. 
 
Notice that the Code column contains values that have leading zeroes.  By using an ODS style override you can 
retain the leading zeros when importing the Code data into Excel.  Microsoft Word will retain the leading zeroes by 
default.  
 

ODS BASICS 
 
ODS is part of BASE SAS software and enables you to generate different types of output from your procedure code.  
An ODS tagset controls the type of output generated (HTML, RTF, PDF, etc.) while an ODS style controls the 
appearance of the output.  We will be using a tagset that creates HTML output that can be opened with Excel and 
Word and also a tagset that creates XML that can be opened by Excel.  The latter tagset is capable of creating an 
Excel workbook with multiple worksheets. 
 
XML is an acronym for Extensible Markup Language, and represents a way to define and format data for easy 
exchange. XML is similar to HTML in that it uses tags. Unlike HTML, whose tags control how data is rendered, XML 
tags describe the structure and meaning of data but do not control how it is rendered.  
 
The ODS statements that generate content compatible with Office follow this general form: 
 

 ods listing close; 
 ods TagsetName style=StyleName file= ... 

    *  your SAS code here; 
 ods TagsetName close; 

 
The ODS statement at  turns off the standard "line printer" ODS destination.  We are only concerned with 
generating HTML and XML output for use with Excel and Word. 
 
The ODS statement at  generates the HTML or XML output, depending on the tagset you use, and stores the 
output in a file of your choosing.  The STYLE option controls the appearance of the output, such as the font and 
color scheme.  To see a list of ODS styles that are available for use at your installation, submit the following SAS 
code: 
 

ods listing; 
proc template; list styles; run; quit; 

 
The ODS statement at  closes and releases the HTML or XML file so that it can be opened with Excel or Word. 

Later in this paper we will discuss using both the ODS tagset named MSOffice2K to create HTML output and the 
ExcelXP tagset to create XML output from the PRINT and TABULATE procedures.  In both cases, appearance 
characteristics such as colors and fonts will be controlled by the ODS style named Banker. 
 
Although you can store your output on a local disk (where SAS software is installed), on a network-accessible disk, 
or a Web server, there are good reasons to store your SAS output on a Web server: 
 

1. The files are available to anyone with network access. 
2. The HTML and XML files can be accessed by a Web browser in addition to Excel and Word. 
3. You can take advantage of Web server authentication and security models. 

 
If you place the files where others can access them over a network, you should set file permissions to prevent 
accidental alteration. 
 

OPENING ODS OUTPUT WITH EXCEL AND WORD 
 
To open an ODS-generated file stored on a Web server, follow these steps: 
 

1. In Excel or Word select File ➨ Open… 
2. In the "File name" field, specify 

  http://Web-server/directory/  
where Web-server corresponds to the domain name for your Web server.  Click Open.  You should see a 
list of files available on your Web server. 

3. Select the file you wish to open and click Open to import the HTML or XML file. 
 
If you don't see a list of available files after step 2, you may have to configure your Web server to allow directory 
listing.  Alternatively, you can specify the full path to the file you wish to open: 
 

http://Web-server/directory/filename 
 
To open ODS-generated files from a local or network accessible disk, follow the same steps, except in step 2 you 
should either navigate to the desired file or type the file name and path in the "File name" field. 
 
Excel and Word will read and convert the HTML or XML file (Excel only) to their respective native format.  After that, 
you can perform any Excel or Word function on the data. 
 

USING ODS TO GENERATE HTML FOR EXCEL AND WORD 
 
The following SAS code is used to generate HTML output for use with Excel and Word: 
 

ods listing close; 
ods tagsets.MSOffice2K style=Banker file='aedata.htm'; 
  title; footnote; 
  proc print data=pharma.phcae noobs label; 
    by protocol; 
    var patient visit aedate aecode aetext aesev aesevc frequency; 
  run; quit; 
 
  proc tabulate data=pharma.phcae; 
    by protocol; 
    var aesev; 
    class aetext aesevc; 
    table aetext*aesevc,aesev*pctn; 
    keyword all pctn; 
    keylabel pctn='Percent'; 
  run; quit; 
ods tagsets.MSOffice2K close; 

 
Figure 3.  SAS code to generate HTML output. 

The code above uses the MSOffice2K tagset to generate the HTML file, and the Banker style to control the 
appearance of the HTML.  Figure 4 shows the resulting HTML file opened in Word. 
 

  
 
 
 
 
 
 
 
 
 
Figure 4.  ODS HTML output viewed with Microsoft Word. 
 
Word does a good job of preserving the format and nature of the SAS output.  The results shown in Figure 4 very 
closely match what you would see if you opened the HTML file with a Web browser.  Note that all the SAS output is 
in a single document; Figure 4 is showing two different views of the single document. 
 
However, opening the same HTML output in Excel (Figure 5) reveals that there are some formatting and data issues. 

Figure 5.  ODS HTML output viewed with Microsoft Excel. 
 
The background color for some of the cells is gray instead of green, the dates are formatted differently than they 
were in Word, and the leading zeros in the adverse event code have been dropped.  The cause and a means of 
correcting these problems will be discussed in the section "Correcting Common Formatting Problems".  Note that all 
the SAS output is in a single worksheet; Figure 5 merely displays two different views of the single worksheet. 
 

USING ODS TO GENERATE XML FOR EXCEL 
 
Up to this point we have used an ODS tagset to create HTML that is imported into Excel, but Excel is also capable of 
reading XML files, and it does an especially good job of handling files that conform to the Microsoft XML Spread-
sheet Specification ("XML Spreadsheet Reference", Microsoft Corp.).  The SAS 9 ExcelXP tagset generates this 
form of XML for use with Microsoft Excel.  The main benefit of using the ExcelXP (XML) tagset is that each table of 
the SAS output is placed into a separate worksheet within the workbook, in contrast to a single worksheet as shown 

in Figure 5 when using the MSOffice2K (HTML) tagset.   However, there are some current disadvantages to using 
the ExcelXP tagset that you should be aware of: 
 

• The tagset is experimental.  This means that limited testing has been performed on it, and its functionality 
may change in the future. 

• The tagset has not yet been optimized.  It may take an unexpectedly long time to generate an XML file. 
• You cannot use graphics in your SAS XML output because Excel does not support embedded graphics in 

an XML file. 
• Because Excel does not automatically resize text-based columns when reading an XML file, when you first 

open the XML file most of the columns are not wide enough to show all the data or headings.  A future 
release of the ExcelXP tagset may correct this problem. 

 
Significant enhancements have been made to the ExcelXP tagset since the initial release of SAS 9.1.  You can 
download a recent version of the ExcelXP tagset from the SAS Presents Web site 
(http://support.sas.com/saspresents).  Look for an entry on that page labeled, "From SAS to Excel via XML".  The 
download contains a file named "excelxp.sas", which contains the SAS code needed to create the ExcelXP tagset.  
Save a copy of this file and submit this SAS code to make the tagset available: 
 

 libname myLib 'directory-for-tagset'; * location to store the tagset; 
 
 ods path myLib.tmplmst(update) sashelp.tmplmst(read); 

 
   %include 'excelxp.sas'; 

 
The LIBNAME statement at  specifies where to store the tagset.  Although you can use the WORK library to 
temporarily store the tagset, a more efficient process would be to create the tagset and store it in a permanent library 
so that you can reference it in other SAS programs.  
 
The ODS PATH statement at  specifies the locations and the order in which to search for ODS tagsets and styles. 
Note that the access mode for the location "myLib.tmplmst" is specified as "update", while "sashelp.tmplmst" is 
specified as "read".  Because ODS searches the PATH in the order given, and because myLib.tmplmst has an 
“update” access mode, PROC TEMPLATE will store the tagset in a file named "tmplmst.sas7bitm" in the directory 
associated with the MYLIB library. 
 
You can use the ExcelXP tagset by specifying the tagset name in an ODS statement and open the resulting XML file 
with Excel.  To continue using the sample code shown in Figure 3, simply change the ODS statement as shown 
below: 
 

ods listing close; 
ods tagsets.ExcelXP style=Banker file='aedata.xml'; 
  *  PROC PRINT and PROC TABULATE code here; 
ods tagsets.ExcelXP close; 

 
Figure 6 (below) shows the results of opening the XML file with Excel.  As was the case with the HTML file, some cell 
background colors appear gray instead of green and the leading zeros in the adverse event code have been 
dropped.  Additionally, border lines are missing from all cells.  Unlike HTML, however, all the tables created by the 
PRINT and TABULATE procedures appear in their own worksheet. 
 
The next section discusses how to correct the formatting issues noted above. 

Figure 6.  ODS XML output viewed with Microsoft Excel, after manually resizing columns. 
 

CORRECTING COMMON FORMATTING PROBLEMS 
 
When Excel and Word import HTML or XML files, the files are converted to native Excel or Word format.  Sometimes 
this conversion process is less than perfect, with Word generally handling the conversion better than Excel.  
Although you can use the features of Excel and Word to manually correct the appearance of the file, doing so will be 
cumbersome if you have a large number of files to correct.  It is much more efficient to have ODS generate a file 
compatible with Office in the first place.  This is accomplished by creating custom ODS styles and by using ODS 
style overrides. 
 

USING STYLE OVERRIDES TO CORRECT OUTPUT-SPECIFIC PROBLEMS 
 
Although ODS styles define a particular appearance for your SAS output, you can change the attributes defined in 
the style at run time by using a "style override".  Style overrides are supported by the PRINT, TABULATE, and 
REPORT procedures, and are best used for changing a small number of features that are associated with a 
particular output.  In our case, style overrides can be used to correct the missing leading zeroes and improperly 
formatted date columns in the HTML and XML files that were imported into Excel. 
 
Upon opening an HTML or XML file, Excel automatically assigns an Excel format based on the data in the cells.  
Sometimes, the default format is not the best choice.  Fortunately, we can use a style override to instruct ODS to use 
a specific Excel format for a column.  Since only the output from the PRINT procedure was affected, the VAR 
statement is the only line of code that needs to be changed. 
 
To correct the date formatting and missing leading zero problems in the HTML file in Figure 5, change the single 
VAR statement in Figure 3 to the following: 

Figure 7.  Default HTML colors supported by Excel 2000 
                and later.

var patient visit; 
var aedate / style={htmlstyle="mso-number-format:ddmmmyyyy"}; 
var aecode / style={htmlstyle="mso-number-format:00000000"}; 
var aetext aesev aesevc frequency; 

 
This code instructs ODS to assign the special Excel format "ddmmmyyyy" for the AEDATE column and the 
"00000000" format for the AECODE column.  The first format is self-explanatory and the second format is 
comparable to the SAS Z8. format.  The HTML file created by ODS can be viewed by Excel, Word or a Web 
browser, since Web browsers ignore the mso-number-format style information when they render the HTML.  
Complete code for correcting the HTML output can be found in the Appendix in the section "Corrected Code to 
Export SAS Output to Excel or Word as HTML". 
 
You can use the same concept to correct the leading zero problem with the XML file shown in Figure 6, but the 
syntax is slightly different: 
 

var patient visit aedate; 
var aecode / style={tagattr="\00000000"}; 
var aetext aesev aesevc frequency; 

 
The next section describes how to correct the remainder of the formatting problems.  Complete code for correcting 
the XML output can be found in the Appendix in the section "Corrected Code to Export SAS Output to Excel as 
XML".  For more information on Excel formats, refer to this author's SUGI 28 paper (DelGobbo, 2003). 
 

USING PROC TEMPLATE TO CORRECT GLOBAL STYLE PROBLEMS 
 
"Global" problems are those that affect all output generated by ODS.  For example, all header and row header cell 
background colors are wrong in all ODS output, as illustrated by Figures 5 and 6.  Since these problems affect all 
output generated by ODS, it is more efficient to make a correction to the ODS style, rather than using an override in 
each of your SAS programs. 
 
Excel mapped the green to gray because the shade of green specified by ODS is not supported by Excel by default.  
If you view the HTML source code and identify the cells that are affected, you will notice that they are assigned the 
Cascading Style Sheet (CSS) class Header or RowHeader.  That is, the HTML contains class="Header" or 
class="RowHeader" in the TH tags.  If you further examine the HTML file, you will see the following CSS definitions 
for the Header and RowHeader classes: 
 

.Header 
{ 
  font-family: 'Times New Roman', Times, serif; 
  font-size: 14pt; 
  font-weight: bold; 
  font-style: normal; 
  color: #033366; 
  background-color: #C4E4B8; 
} 

.RowHeader 
{ 
  font-family: 'Times New Roman', Times, serif; 
  font-size: 14pt; 
  font-weight: bold; 
  font-style: normal; 
  color: #033366; 
  background-color: #C4E4B8; 
} 

 
Both of these classes use the HTML color #C4E4B8.  However, if you refer to Figure 7, you will see that this color is 
not supported by Excel by default.  Thus, Excel maps this color to gray.  Excel does support a shade of green, 
#CCFFCC, which is very close to the original.  
We will next learn how to use the TEMPLATE 
procedure to modify the Banker style to use this 
new shade of green. 
 
The TEMPLATE procedure is part of BASE 
SAS, and can be used to list, create and modify 
ODS styles and tagsets.   Because an in-depth 
discussion of this procedure is beyond the 
scope of this paper, see Chapter 5 of the ODS 
documentation (SAS Institute Inc., 1999) for 
more information on using the TEMPLATE 
procedure. 

Before you can make a change to the style, you must first determine the nature of the change that needs to be 
made.  Submit this SAS code to list the source of the Banker style: 
 

ods listing; 
proc template; 
  source styles.Banker; 
run; quit; 

 
By examining the output, you will see that the following colors are used: 
 

   replace colors / 
      'headerfgemph'   = cx033366 
      'headerbgemph'   = cxFFFFCD 
      'headerfgstrong' = cx033366 
      'headerbgstrong' = cxFFFFCD 
      'headerfg'       = cx033366 
      'headerbg'       = cxC4E4B8 
                   . . . 

 
Note that SAS uses the "cx" notation to represent hexadecimal color values while HTML uses "#".  By comparing the 
above code to Figure 7, you can see that the Banker style is using 3 colors that are not supported by Excel: 
cx033366, cxFFFFCD and cxC4E4B8.  However, Excel maps cx033366 to #003366, which is so close that you will 
probably not notice the difference.  Similarly, cxFFFFCD is mapped to #FFFFCC, which is also a very good match.  
Thus, the only change needed to the Banker style is to substitute cxCCFFCC for cxC4E4B8.  This is accomplished 
with the following code (complete code can be found in the Appendix in the section "Code to Create the MSOBanker 
Style"): 
 

libname myLib 'directory-for-style'; * location to store the style; 
 
ods path myLib.tmplmst(update) sashelp.tmplmst(read); 
 
proc template; 
  define style styles.MSOBanker; 
    parent = styles.Banker; 
    replace colors / 
      'headerfgemph'   = cx033366 
      'headerbgemph'   = cxFFFFCD 
      'headerfgstrong' = cx033366 
      'headerbgstrong' = cxFFFFCD 
      'headerfg'       = cx033366 
      'headerbg'       = cxCCFFCC 
                   . . . 

 
This new style, named MSOBanker, will be stored in myLib.tmplmst and has all the attributes of the parent Banker 
style; the only difference is that the header and row header cell background colors will be #CCFFCC instead of 
#C4E4B8. 
 
You can use the MSOBanker style to create a new HTML file by specifying the style name in an ODS statement and 
open the resulting HTML file with Excel and Word.  To continue using the sample code shown in Figure 3 with the 
style overrides discussed in the previous section, change the ODS statement as shown below: 
 

ods listing close; 
ods tagsets.MSOffice2K style=MSOBanker file='aedata.htm'; 
  *  PROC PRINT code w/ style overrides and PROC TABULATE code here; 
ods tagsets.MSOffice2K close; 

 
Figures 8 and 9 (below) show the results of opening the HTML file with Word and Excel, respectively.  All of the 
formatting problems have been corrected by using a combination of style overrides and the new MSOBanker style. 
Complete code for correcting the HTML output can be found in the Appendix in the section "Corrected Code to 
Export SAS Output to Excel or Word as HTML". 

Figure 8.  Corrected ODS HTML output viewed with Microsoft Word. 

Figure 9.  Corrected ODS HTML output viewed with Microsoft Excel. 
 
We will now focus on making the final corrections to the XML file generated by ODS.  In the previous section we 
discussed using style overrides to handle the missing leading zero problem when the XML file is opened with Excel. 
There are still two outstanding issues with the format of the XML file: the green background is mapped to gray by 
Excel and all cell border lines are missing.  Both of these problems can be solved by creating a new ODS style for 
use with Excel. 
 
If you examine the XML file generated by ODS and identify the cells that have missing border lines, you will see that 
they are using the style classes named "Header", "RowHeader" and "Data".  That is, you will see StyleID="Header", 
StyleID="RowHeader" or StyleID="Data" in the Cell tag.  When you further examine the XML file, you will see the 
following definitions for the Header, RowHeader and Data styles: 

<Style ss:ID="Header"> 
<ss:Borders> 
<ss:Border ss:Position="Left" /> 
<ss:Border ss:Position="Top" /> 
<ss:Border ss:Position="Right" /> 
<ss:Border ss:Position="Bottom" /> 
</ss:Borders> 
<Interior ss:Color="#C4E4B8" 
ss:Pattern="Solid" /> 
  . . .  
</Style> 

<Style ss:ID="RowHeader"> 
<ss:Borders> 
<ss:Border ss:Position="Left" /> 
<ss:Border ss:Position="Top" /> 
<ss:Border ss:Position="Right" /> 
<ss:Border ss:Position="Bottom" /> 
</ss:Borders> 
<Interior ss:Color="#C4E4B8" 
ss:Pattern="Solid" /> 
  . . .  
</Style> 

<Style ss:ID="Data"> 
<ss:Borders> 
<ss:Border ss:Position="Left" /> 
<ss:Border ss:Position="Top" /> 
<ss:Border ss:Position="Right" /> 
<ss:Border ss:Position="Bottom" /> 
</ss:Borders> 
<Interior ss:Color="#FFFFCD" 
ss:Pattern="Solid" /> 
  . . .  
</Style> 

 
The Header and RowHeader styles are using the unsupported green color for the background color of the cells.  This 
can be corrected by using the MSOBanker style that you made earlier.  The Border tags of each style have only a 
Position attribute, and lack any attributes to indicate the thickness or other properties of the border lines.  You could 
modify the MSOBanker style to change the border width, but that would influence the already good-looking border 
lines in the HTML files.  The best solution is to create a new style for use with the ExcelXP tagset: 
 

libname myLib 'directory-for-style'; * location to store the style; 
 
ods path myLib.tmplmst(update) sashelp.tmplmst(read); 
 
proc template; 
  define style styles.XLBanker; 
    parent = styles.MSOBanker; 
    style Header from Header / 
      borderwidth=2; 
    style RowHeader from RowHeader / 
      borderwidth=2; 
    style Data from Data / 
      borderwidth=2; 
  end; 
run; quit; 

 
This new style, named XLBanker, will be stored in myLib.tmplmst and has all the attributes of the parent MSOBanker 
style that was created earlier; the only difference is that border line widths were adjusted for use with Excel XML. 
 
You can use the XLBanker style to create a new XML file by specifying the style name in an ODS statement and 
open the resulting XML file with Excel.  To continue using the sample code shown in Figure 3 with the style overrides 
discussed in the previous section, change the ODS statement as shown below: 
 

ods listing close; 
ods tagsets.ExcelXP style=XLBanker file='aedata.xml'; 
  *  PROC PRINT code w/ style overrides and PROC TABULATE code here; 
ods tagsets.ExcelXP close; 

 
Figure 10 (below) shows the results of opening the XML file with Excel.  All of the formatting problems have been 
corrected by using a combination of style overrides and the new XLBanker style.  Complete code for correcting the 
XML output can be found in the Appendix in the section "Corrected Code to Export SAS Output to Excel as XML". 

Figure 10.  Corrected ODS XML output viewed with Microsoft Excel, after manually resizing columns. 
 

MOVING EXCEL DATA TO SAS 
 
Figure 11 (below) shows the Excel workbook we want to import into SAS. The workbook contains 4 worksheets. We 
want to import the data from each worksheet into a different SAS table, using the name of the worksheet for the table 
name. The sections that follow explain how to save the workbook as an XML file, and then how to load the workbook 
into SAS using a provided SAS macro and SAS XMLMap, both designed specifically for Excel XML data. All the 
necessary SAS code is supplied for you. 
 
To download the sample data, SAS macro and SAS XMLMap, go to the SAS Presents Web site 
(http://support.sas.com/saspresents) and look for an entry on that page labeled, "From SAS to Excel via XML". There 
you will find a download package with all the files you need. 

Figure 11.  Excel workbook with data to import into SAS. 
 

SAVING AN EXCEL WORKBOOK AS XML 
 
You must save an Excel workbook as an XML file before you can import it into SAS.  To do so, in Excel select  
File ➨ Save As, and choose "XML Spreadsheet".  Specify a name and directory for the file.  For the purpose of this 
paper, we saved the XML in a file named "mydata.xml". 
 
To ensure that SAS can access the mydata.xml file, you can save the file on a network-accessible drive.  Then you 
can access the file as if it were native to the operating system where SAS is installed.  Another solution is to save the 
file to a location that is under the control of a Web server.  SAS will then be able to read the file using the URL 
access method, which is part of BASE SAS.  These techniques are useful if SAS is installed on a different machine 
from Excel.  If neither of these options is available to you, using FTP or some other method, move the file from the 
Windows machine to the machine that has SAS installed. 
  
If you place the files where others can access them, be sure to set file permissions to prevent accidental alteration. 
 

AUTOMATIC DATA CONVERSION BETWEEN EXCEL AND SAS 
 
You need to be aware of some of the behaviors to expect when converting Excel data to SAS data.  As mentioned 
earlier, when the worksheets in Figure 11 are imported into SAS tables, the name of the worksheet, "Chemicals" (for 
example) is used as the SAS table name.  However, the technique of using the worksheet name for the table name 
does not always work.  For example, a worksheet named "40-107 Senate Voting" cannot be used for the SAS table 
name because it begins with a number and contains invalid characters (the spaces and a dash).  Therefore, the 
provided SAS code that loads the worksheet into SAS must convert the "4" and other invalid characters to an 
underscore ("_").  In this case the resulting SAS table name would be "_0_107_Senate Voting".  Likewise, importing 
a worksheet named "Health & Wellness Resources" would result in a table named "Health___wellness_resources". 
 
Attempting to use Excel column labels for SAS table column names can result in a similar problem.  Note that none 
of the column labels in Figure 11 can be used as SAS column names because they all contain invalid characters 
such as blanks, parentheses and the forward slash ("/").  Additionally, the first column does not have a label.  The 
SAS code that loads the worksheet into SAS must create valid column names by converting invalid characters to an 
underscore ("_").  SAS column labels are set to the original value of the Excel label.  The table below shows 
examples of how some of the Excel column labels in Figure 11 will be converted to SAS column names. 

SAS Table Name SAS Label 
Chemicals Chemicals 
Graphics_cards Graphics Cards 
Health___wellness_resources Health & Wellness Resources 
_0_107_senate_voting 40-107 Senate Voting 

 
Table 2.  SAS tables created from Excel workbook. 

Excel Column Label SAS Column Name SAS Column Label 
(blank) A unique 32-character name starting with "_" . 
Pixel pipelines Pixel_pipelines Pixel pipelines 
Peak fill rate (Mpixels/s) Peak_fill_rate__Mpixels_s_ Peak fill rate (Mpixels/s) 

 
        Table 1.  Conversion of Excel column labels to SAS column names. 
 

THE SAS XML LIBNAME ENGINE (SXLE) AND XMLMAPS 
 
The SAS XML LIBNAME Engine can import an XML file into a SAS table or export a SAS table as an XML file.  Thus, 
you can use the SXLE to exchange data between SAS and third-party, XML-aware applications such as Excel. 
 
Although the SXLE has been available since SAS release 8.1, recent improvements have made it possible to 
precisely control how data is imported.  The new SAS XMLMap enables you to map any XML element or attribute to 
a column or row in a SAS table.  The SXLE then uses the XMLMap to control how the XML data is imported into a 
SAS table.  You can manually create the XMLMap using a text editor (this is not recommended), or you can use the 
new XML Mapper (formerly know as XML Atlas) that provides a point-and-click interface. 
 

READING EXCEL XML INTO SAS 
 
You could use the SAS Libname Engine and a SAS XMLMap of your own creation to import an Excel XML file into 
SAS, but because the Excel XML format and the data conversion issues are quite complex, SAS provides an 
XMLMap and corresponding SAS code specific to Excel that loads the Excel XML into SAS tables.  As mentioned 
earlier, both components as well as a SAS macro to import the XML data, are available for download from the SAS 
Presents Web site (http://support.sas.com/saspresents).  These components may ship with future releases of SAS, 
so watch the Base SAS Community Web site for further details (http://support.sas.com/rnd/base/) 
 
Download the XMLMap and the SAS macro and make the two files available on the platform where SAS is installed.  
This paper assumes that you saved the SAS XMLMap in a file named "excelxp.map" and the SAS macro to load the 
XML data was stored in a file named "loadxl.sas".  The file "loadxl.sas" contains a SAS macro named XLXP2SAS, 
which is used to import the Excel XML file into SAS tables. 
 
The easiest way to explain how to use the macro is with a few examples.  First, we will consider importing the XML 
workbook shown in Figure 11 when either SAS is installed on the same machine that has the XML file or SAS is 
installed on a different machine or platform, but the XML file is accessible via a network drive.  To import the 
workbook into SAS, submit this code, making sure to include the appropriate directory paths: 
 

 %include 'loadxl.sas'; 
 
 %xlxp2sas(excelfile=mydata.xml,  

             mapfile=excelxp.map); 
 
The statement at  makes the XLXP2SAS macro available to SAS.  The statement at  imports the data from all 
the worksheets into separate SAS tables.  By default the SAS tables are created in the WORK library.  You can 
control the library used to store the SAS tables by specifying the LIBRARY argument of the XLXP2SAS macro.  For 
example, to store the tables in the SASUSER library, submit this code: 
 

%xlxp2sas(excelfile=mydata.xml,  
          mapfile=excelxp.map, 
          library=sasuser); 

 
 
Table 2 lists the SAS tables 
created as a result of importing 
the Excel workbook shown in 
Figure 11.  Note that while the 
table names may look a bit odd, 
the actual worksheet names are 
used in the SAS labels. 

Figure 12 shows a portion of the "Graphics_cards" table.  By comparing Figures 11 and 12, you can see that the 
XLXP2SAS macro successfully imported the "Graphics Cards" worksheet as a SAS table.  For instance, the columns 
"Pixel pipelines" and "Memory bus width (bits)" were both correctly typed as character, because those columns 
contain data such as "2*2" and "128*2", respectively.  
 

 
 
Figure 12.  SAS Graphics_cards table, created by importing an Excel workbook. 
 
Up to this point the assumption is that the XML file resides on the same machine as SAS or was available via a 
network drive.  However, if the XML file resides on the Web server of a remote machine, you can use the URL 
access method to retrieve the file by submitting the following code: 
 

filename myxml URL 'http://Web-server/mydata.xml'; 
%xlxp2sas(excelfile=FILEREF:myxml,  
          mapfile=excelxp.map); 

 
The submitted code causes XLXP2SAS to contact the Web server to retrieve the XML file, rather than looking for it 
on a local disk.  The FILEREF modifier can also be used with the MAPFILE argument to retrieve the SAS XMLMap 
from a Web server. 
 
Documentation for the XLXP2SAS macro can be found in the Appendix of this paper. 
 

XLXP2SAS DRAWBACKS AND LIMITATIONS 
 
There are a few issues surrounding XLXP2SAS that you should be aware of: 
 

1. XLXP2SAS creates temporary tables in the WORK library.  These tables can get very large.  However, this 
is generally not a problem unless your system is very low in disk space, as the temporary files are 
automatically cleaned up after XLXP2SAS runs (unless the argument CLEANUP=N was specified). 

2. The data in every worksheet you want to import must be fairly rectangular.  Although the XLXP2SAS macro 
attempts to handle non-rectangular data by adding missing values, the results can be unpredictable if the 
data is too sparse. 

3. By default, XLXP2SAS runs with the argument HASLABELS=Y.  This implies that all worksheets in a 
workbook have column labels in the first row.  If none of your worksheets contain column labels in the first 
row, specify HASLABELS=N when you invoke XLXP2SAS.  The HASLABELS argument applies to all 
worksheets in a workbook. 

4. If you specify HASLABELS=N, the column names in the SAS table(s) will be of the form "COLUMN1", 
"COLUMN2", "COLUMN3" and so on, and the respective column labels will be "Column 1", "Column 2" and 
"Column 3". 

SAS SERVER TECHNOLOGY 
 
If you have licensed SAS/IntrNet® software, you can dynamically incorporate SAS output into Excel using the 
Application Dispatcher.  You can perform similar tasks with the Stored Process Server, which is new for SAS 9.1. 
 
The Application Dispatcher and the Stored Process Server enable you to execute SAS programs from a Web 
browser or any other client that can open an HTTP connection to either of these SAS servers (which can run on any 
platform where SAS is licensed).  The SAS programs that you execute from the browser can consist of any 
combination of DATA step, PROC, MACRO, or SCL code.  Thus, all of the code shown up to this point can be 
executed using either Application Dispatcher or the Stored Process Server. 
 
Detailed information on using these products to facilitate data interchange between SAS and Microsoft Office is 
presented elsewhere (DelGobbo, 2002; DelGobbo 2003). 
 

CONCLUSION 
 
Using ODS to generate HTML and XML output is an effective method of incorporating SAS output in Excel and Word 
documents.  Although you may initially encounter formatting problems when using this technique, by using ODS 
style overrides and the TEMPLATE procedure you can correct these problems.  
 
The SAS 9.1 ExcelXP tagset complies with the Microsoft XML Spreadsheet Specification and provides an easy way 
to export your data to Excel workbooks that contain multiple worksheets.  While it may be a bit cumbersome to use 
the SXLE and XMLMaps to import Excel data to SAS, the use of the XLXP2SAS macro greatly simplifies the task. 
 
SAS Institute continues to work toward better Microsoft Office integration, and future releases of SAS software will 
provide even more robust means of using SAS content with Office applications. 
 

APPENDIX 
 

CODE TO CREATE THE MSOBANKER STYLE 
 

libname myLib 'directory-for-style'; * location to store the style; 
 
ods path myLib.tmplmst(update) sashelp.tmplmst(read); 
 
proc template; 
  define style styles.MSOBanker; 
    parent = styles.Banker; 
    replace colors / 
      'headerfgemph'   = cx033366 
      'headerbgemph'   = cxFFFFCD 
      'headerfgstrong' = cx033366 
      'headerbgstrong' = cxFFFFCD 
      'headerfg'       = cx033366 
      'headerbg'       = cxCCFFCC 
      'datafgemph'     = cx000000 
      'databgemph'     = cxFFFFCD 
      'datafgstrong'   = cx000000 
      'databgstrong'   = cxFFFFCD 
      'datafg'         = cx000000 
      'databg'         = cxFFFFCD 
      'batchfg'        = cx000000 
      'batchbg'        = cxFFFFCD 
      'tableborder'    = cx000000 
      'tablebg'        = cx000000 
      'notefg'         = cx033366 
      'notebg'         = cxFFFFCD 
      'bylinefg'       = cx033366 
      'bylinebg'       = cxFFFFCD 
      'captionfg'      = cx033366 
      'captionbg'      = cxFFFFCD 

      'proctitlefg'    = cx033366 
      'proctitlebg'    = cxFFFFCD 
      'titlefg'        = cx033366 
      'titlebg'        = cxFFFFCD 
      'systitlefg'     = cx033366 
      'systitlebg'     = cxFFFFCD 
      'Conentryfg'     = cx033366 
      'Confolderfg'    = cx033366 
      'Contitlefg'     = cx033366 
      'link2'          = cx800080 
      'link1'          = cx0000FF 
      'contentfg'      = cx000000 
      'contentbg'      = cxFFFFCD 
      'docfg'          = cx033366 
      'docbg'          = cxFFFFCD; 
  end; 
run; quit; 

 
Note that cx000000 is black, cx800080 is purple and cx0000FF is blue.  According to Figure 7, these colors are 
supported by Excel by default, so no remapping is needed for these colors. 
 

CORRECTED CODE TO EXPORT SAS OUTPUT TO EXCEL OR WORD AS HTML 
 

ods listing close; 
ods tagsets.MSOffice2K style=MSOBanker file='aedata.htm'; 
  title; footnote; 
  proc print data=pharma.phcae noobs label; 
    by protocol; 
    var patient visit; 
    var aedate / style={htmlstyle="mso-number-format:ddmmmyyyy"}; 
    var aecode / style={htmlstyle="mso-number-format:00000000"}; 
    var aetext aesev aesevc frequency; 
  run; quit; 
 
  proc tabulate data=pharma.phcae; 
    by protocol; 
    var aesev; 
    class aetext aesevc; 
    table aetext*aesevc,aesev*pctn; 
    keyword all pctn; 
    keylabel pctn='Percent'; 
  run; quit; 
ods tagsets.MSOffice2K close; 

 

CORRECTED CODE TO EXPORT SAS OUTPUT TO EXCEL AS XML 
 

ods listing close; 
ods tagsets.ExcelXP style=XLBanker file='aedata.xml'; 
  title; footnote; 
  proc print data=pharma.phcae noobs label; 
    by protocol; 
    var patient visit aedate; 
    var aecode / style={tagattr="\00000000"}; 
    var aetext aesev aesevc frequency; 
  run; quit; 
 
  proc tabulate data=pharma.phcae; 
    by protocol; 
    var aesev; 
    class aetext aesevc; 
    table aetext*aesevc,aesev*pctn; 
    keyword all pctn; 
    keylabel pctn='Percent'; 

  run; quit; 
ods tagsets.ExcelXP close; 

 

XLXP2SAS MACRO DOCUMENTATION 
 
Below is a list of the arguments supported by the XLXP2SAS macro.  All arguments require a value, except where a 
default value is indicated. 
 

 
Argument 

 
Description 

Default 
Value 

EXCELFILE The name and path for the Excel XML file you want to import to SAS.  Do not use 
quotes in this value.  To specify a SAS FILEREF instead of a file, use 
FILEREF:fref, where fref is the FILEREF. 

 

MAPFILE The name and path of the SAS XMLMap for reading Excel XML files.  Do not use 
quotes in this value.  To specify a SAS FILEREF instead of a file, use 
FILEREF:fref, where fref is the FILEREF.  You can download a copy of the SAS-
provided XMLMap from the SAS Presents Web site 
(http://support.sas.com/saspresents).  Look for an entry on that page labeled 
"From SAS to Excel via XML". 

 

LIBRARY The name of the SAS library where imported tables are stored. WORK 
HASLABELS Controls whether or not the worksheets have column labels in the first row of the 

Excel table.  This setting applies to all worksheets in a workbook.  If set to Y, the 
labels will be used for the SAS column names and labels.  If your workbook does 
not have column labels in the first row of the Excel table, specify N. 

Y 

CLEANUP Controls whether or not to delete temporary SAS files and also whether or not to 
deassign FILEREFs that were used when importing the Excel data to SAS.  
FILEREFS that you explicitly assign with a FILEREF statement will not be 
deassigned.  To disable this feature, specify N. 

Y 

VERBOSE Controls the level of debugging information written to the SAS LOG.  Specify Y to 
activate this feature. 

N 

